

Construção de Compiladores
Daniel Lucrédio, Helena Caseli, Mário César San Felice e Murilo Naldi
Tópico 03 - Análise Sintática Introdução - Lista de Exercícios Resolvida
(Última revisão: jan/2020)

1. Qual é a função da etapa de análise sintática no processo de tradução de um compilador?

R: Compreender a estrutura de um programa fonte. Procura detectar se a forma com que as unidades léxicas (tokens) se compõem está correta. Olha apenas a forma, e não o significado.

2. Faça a análise sintática do seguinte ditado: "A fé move montanhas" (pesquise um pouco de gramática da língua portuguesa)

R:



3. Qual a importância da teoria da computação (linguagens e autômatos) para a análise sintática?

R: Definir uma forma mais eficiente para descrever linguagens (gramáticas livres de contexto) e implementar parsers ou analisadores sintáticos (autômatos com pilha). Sem esse formalismo, o trabalho deve ser feito de forma manual, o que torna mais difícil a criação e manutenção dos compiladores.

4. Porque utiliza-se gramáticas livres de contexto em compiladores, para análise sintática, e não outras classes de gramáticas?

R: Porque elas são simples o suficiente para serem concebidas e manipuladas. São adequadas para descrever a maioria das construções das linguagens utilizadas na prática. Além disso, permitem fácil implementação (autômatos com pilha). Outras classes não são adequadas. Linguagens regulares são simples demais, não permitindo recursividade. Linguagens sensíveis a contexto ou recursivamente enumeráveis são de difícil implementação (máquina de turing).

5. Qual a diferença entre uma árvore de análise sintática e uma árvore de sintaxe abstrata?

R: A árvore de análise sintática representa exatamente o processo de análise. Cada nó-folha da árvore é um terminal da gramática, e cada nó-interior é um não-terminal da gramática. O mapeamento é um-para-um. Já a árvore de sintaxe abstrata contém somente as informações necessárias para a análise semântica.

6. Defina uma gramática não ambígua para expressões aritméticas onde os operandos são constantes numéricas e variáveis. Os operadores são os operadores aritméticos comuns: *,/,+,-, além do exponencial ^. A precedência e associatividade dos operadores, porém, é o contrário do usual, ou seja: + e - tem precedência sobre * e / que tem precedência sobre ^. Todos os operadores são associativos à direita, com exceção do exponencial "^" que é associativo à esquerda.

R:

```
expr : expr op1 termo | termo;
termo : fator op2 termo | fator;
fator : elemento op3 fator | elemento;
elemento: NUM | VAR;
op1 : '^';
op2: '*' | '/';
op3: '+' | '-';
```

7. Modifique a gramática do exercício anterior de forma que expoentes só possam envolver constantes numéricas, e nunca variáveis

```
expr : expr op1 termoSemVar | termo;
termo : fator op2 termo | fator;
fator : elemento op3 fator | elemento;
elemento: NUM | VAR;
termoSemVar : fatorSemVar op2 termoSemVar | fatorSemVar;
fatorSemVar : elementoSemVar op3 fatorSemVar | elementoSemVar;
elementoSemVar: NUM;
op1 : '^';
op2: '*' | '/';
op3: '+' | '-';
```

8. Considere a seguinte gramática

E : E op E | (E) | NUM

Modifique essa gramática para que ela passe a utilizar uma notação pré-fixada, ou seja, ao invés de escrever $2 + 3 * 4$, escreveríamos $+ 2 * 3 4$. A gramática resultante precisa de parêntesis? A gramática resultante é ambígua? Justifique suas respostas.

R:

E : op E E | NUM

Não é mais necessário parêntesis, pois agora é possível agrupar as expressões simplesmente pela ordem dos operadores. Também não há mais ambiguidade, pois não há recursividade dupla.

9. Considere as seguintes regras gramaticais:

```
comando : comando ';' listaComandos
comando : comando ';' 'if' '(' expr ')' 'then' comando
expr : expr '+' termo | expr '-' termo | termo
termo : termo '*' fator | termo '/' fator | fator
```

```
fator : fator '.' VAR | VAR
comando : 'while' '(' expr ')' comando 'endwhile'
listacomandos : '{' comando '}'
```

Elimine toda recursividade à esquerda

R:

```
comando : 'while' '(' expr ')' comando 'endwhile' comando2
comando2 : ';' listaComandos comando2 | ';' 'if' '(' expr ')' 'then'
          comando | <<vazio>>
expr : termo expr2
expr2 : '+' termo expr2 | '-' termo expr2 | <<vazio>>
termo : fator termo2
termo2 : '*' fator termo2 | '/' fator termo2 | <<vazio>>
fator : VAR fator2
fator2 : '.' VAR fator2 | <<vazio>>
listacomandos : '{' comando '}'
```

10. Considere as seguintes regras gramaticais

```
comandoCondicao
: 'SE' expressaoRelacional 'ENTAO' comando 'FIMSE'
| 'SE' expressaoRelacional 'ENTAO' comando 'SENAO' comando 'FIMSE'
;
```

Fatore à esquerda, caso necessário.

R:

```
comandoCondicao : 'SE' expressaoRelacional 'ENTAO' comando
                 comandoSenao 'FIMSE'
comandoSenao : 'SENAO' comando | <<vazio>>
```

11. Considere as seguintes regras gramaticais

```
lexp : atomo | lista
atomo : numero | identificador
lista : '(' lexpseq ')'
lexpseq : lexpseq lexp | lexp
```

Remova a recursividade à esquerda e modifique as regras de forma a empregar EBNF

R:

```
lexp : atomo | lista
atomo : numero | identificador
lista : '(' lexp+ ')'
```

12. Dada a gramática a seguir, elimine a recursividade à esquerda

```
S : ( L ) | a
L : L , S | S
```

R:

S : '(' L ')' | 'a'

L : S L2

L2 : ',' S L2 | <<vazio>>